Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematical Sciences
goutzicj@oneonta.edu

Lesson 3

Antiderivatives, Direction Fields and Integral Curves

Initializations

```
> restart;
  with(DEtools):
>
```

3.1 Antiderivatives

Examples

Example 3.1.1

Let

$$f(x) = 5x^2 - 7\sqrt{x} + 1$$

Find all antiderivatives of *f*.

Solution
Define the function, integrate and add a constant *c*.

```
> f:=5*x^2-7*sqrt(x)+1;
```

$$f := 5\,x^2 - 7\sqrt{x} + 1 \qquad (2.1.1.1)$$

```
> F:=int(f, x)+c;
```

$$F := \frac{5}{3}\,x^3 - \frac{14}{3}\,x^{3/2} + x + c \qquad (2.1.1.2)$$

```
>
```

Example 3.1.2

Let

$$f''(x) = -3x^2 + 6x - 4$$

while $f(1) = 3$ and $f'(1) = -2$. Compute $f(x)$.

Solution
Define the second derivative and integrate.

```
> fpp:=-3*x^3+6*x-4;
```

$$fpp := -3\,x^3 + 6\,x - 4 \qquad (2.1.2.1)$$

```
> fp:=int(fpp, x)+c[1];
```

$$fp := -\frac{3}{4}\,x^4 + 3\,x^2 - 4\,x + c_1 \qquad (2.1.2.2)$$

Implement the condition $f'(1) = -2$ and solve for c_1.

```
> eq1:=subs(x=1, fp)=-2;
```

$$eq1 := -\frac{7}{4} + c_1 = -2 \qquad (2.1.2.3)$$

```
> val_c[1]:=solve(eq1, c[1]);
```

$$val\_c_1 := -\frac{1}{4} \qquad (2.1.2.4)$$

Update the value of **fp**.

```
> fp:=subs(c[1]=val_c[1], fp);
```

$$fp := -\frac{3}{4}\,x^4 + 3\,x^2 - 4\,x - \frac{1}{4} \qquad (2.1.2.5)$$

Repeat this procedure to obtain the function *f*.

```
> f:=int(fp, x)+c[2];
```

$$f := -\frac{3}{20}\,x^5 + x^3 - 2\,x^2 - \frac{1}{4}\,x + c_2 \qquad (2.1.2.6)$$

```
> eq2:=subs(x=1, f)=3;
```

$$eq2 := -\frac{7}{5} + c_2 = 3 \qquad (2.1.2.7)$$

```
> val_c[2]:=solve(eq2, c[2]);
```

$$val\_c_2 := \frac{22}{5} \qquad (2.1.2.8)$$

```
> f:=subs(c[2]=val_c[2], f);
```

$$f := -\frac{3}{20}\,x^5 + x^3 - 2\,x^2 - \frac{1}{4}\,x + \frac{22}{5} \qquad (2.1.2.9)$$

```
>
```

Example 3.1.3

A ball is thrown upward at time $t = 0$ with a speed of 50 ft/sec from the edge of a cliff 750 ft above the ground. Neglecting air resistance, do the following

i) Find the height of the ball above the ground *t* seconds later.
ii) Compute the time when the ball reaches maximum height.
iii) Find the maximum height reached by the ball.
iv) Compute the speed with which the ball hits the ground.

Solution
i) Find the height of the ball above the ground *t* seconds later.

Let $v(t)$ denote the velocity of the ball at time *t*. Acceleration due to gravity equals 32 ft/sec^2. This implies that

$$\frac{dv}{dt} = -32$$

We integrate this equation and impose the initial condition $v(0) = 50$.

```
> e1:=diff(v(t), t)=-32;
```

(2.1.3.1)

$$e1 := \frac{d}{dt}\, v(t) = -32 \qquad (2.1.3.1)$$

```
> e2:=map(int, e1, t)+(0=c);
```

$$e2 := v(t) = -32\, t + c \qquad (2.1.3.2)$$

```
> eq_c:=subs({t=0, v(t)=50}, e2);
```

$$eq\_c := 50 = c \qquad (2.1.3.3)$$

```
> val_c:=isolate(eq_c, c);
```

$$val\_c := c = 50 \qquad (2.1.3.4)$$

```
> e3:=subs(val_c, e2);
```

$$e3 := v(t) = -32\, t + 50 \qquad (2.1.3.5)$$

If $x(t)$ denotes the height of the ball above the ground at time t, then

$$\frac{dx}{dt} = v(t) \quad \text{and} \quad x(0) = 750$$

Hence, we can repeat the procedure above and find

```
> e4:=subs(v(t)=diff(x(t), t), e3);
```

$$e4 := \frac{d}{dt}\, x(t) = -32\, t + 50 \qquad (2.1.3.6)$$

```
> e5:=map(int, e4, t)+(0=c);
```

$$e5 := x(t) = -16\, t^2 + 50\, t + c \qquad (2.1.3.7)$$

```
> eq_c:=subs({t=0, x(t)=570}, e5);
```

$$eq\_c := 570 = c \qquad (2.1.3.8)$$

```
> val_c:=isolate(eq_c, c);
```

$$val\_c := c = 570 \qquad (2.1.3.9)$$

```
> e6:=subs(val_c, e5);
```

$$e6 := x(t) = -16\, t^2 + 50\, t + 570 \qquad (2.1.3.10)$$

ii) Compute the time when the ball reaches maximum height.

The maximum height is reached if $v(t) = 0$.

```
> eq_t:=rhs(e3)=0;
```

$$eq\_t := -32\, t + 50 = 0 \qquad (2.1.3.11)$$

```
> tmax:=solve(eq_t, t);
```

$$tmax := \frac{25}{16} \qquad (2.1.3.12)$$

iii) Find the maximum height reached by the ball.

The maximum height reached by the ball equals $x\left(\frac{25}{26}\right)$.

```
> xmax:=subs(t=tmax, e6);
  evalf(xmax);
```

$$xmax := x\left(\frac{25}{16}\right) = \frac{9745}{16}$$

$$x\left(\frac{25}{16}\right) = 609.0625000 \quad (2.1.3.13)$$

iv) Compute the speed with which the ball hits the ground.
In order to compute the speed of impact with the ground, we first need to know when the ball hits the ground. That event happens when $x(t) = 0$.

```
> t_impact:=fsolve(subs(x(t)=0, e6), t);
```

$$t\_impact := -4.607297910, 7.732297910 \quad (2.1.3.14)$$

Observe that the first value is negative and can be discarded. (why?)

```
> t_impact:=t_impact[2];
```

$$t\_impact := 7.732297910 \quad (2.1.3.15)$$

The velocity of the ball at time of impact equals $v(7.732297910)$.

```
> v_impact:=subs(t=t_impact, e3);
```

$$v\_impact := v(7.732297910) = -197.4335331 \quad (2.1.3.16)$$

```
>
```

This is measured in ft/sec.

3.2 Direction Fields and Integral Curves.

Examples

Example 3.2.1

Let

$$\boldsymbol{f}(\boldsymbol{x}) = \sqrt{\mathbf{1} + \boldsymbol{x}^3} - \boldsymbol{x}$$

Plot a direction field and an antiderivative F which satisfies $F(-1) = 0$.
(this is example 6 on page 356 of the textbook)

Solution
First, clear the variable **F** which was used in Example 3.1.1.

```
> F:='F';
```

$$F := F \quad (3.1.1.1)$$

Setup the differential equation for F.

$$\frac{dF}{dx} = \sqrt{1 + x^3} - x$$

```
> deq:=diff(F(x), x)=sqrt(1+x^3)-x;
```

$$deq := \frac{d}{dx} F(x) = \sqrt{1 + x^3} - x \quad (3.1.1.2)$$

Code the initial condition.

```
> ic:=[[-1, 0]];
```

$$ic := [[-1, 0]] \quad (3.1.1.3)$$

Use Maple's DEplot routine to visualize the direction field and the antiderivative F that satisfies the initial condition

$$F(-1) = 0$$

```
> DEplot(deq, F(x), x=-1..3, ic, arrows=SLIM, F=-1..4,
  linecolor=blue, scaling=constrained);
```



Please compare this result to Figure 6 on page 356 of the textbook.

>